
Basic polymorphic typechecking

1 Introduction

This handout presents the core of the ML polymorphic type system, and two algorithms for the
system.

2 The basic ML type system

The core ideas of the ML type system can be presented in terms of a simple extendedλ-calculus:

e ::= x variable
| λ x.e λ abstraction
| (e e′) application
| let x = e in e′ let binding

The set of types (τ ∈ TY) is defined by:

τ ::= ι type constant
| α type variable
| (τ1 → τ2) function type

and the set oftype schemes(σ ∈ TYSCHEME) is defined by:

σ ::= τ
| ∀α.σ

The type schemaσ = ∀α1.∀α2 · · · ∀αn.τ is abbreviated as∀α1α2 · · ·αn.τ . The type variables
α1, . . . , αn are said to beboundin σ. A type variable that occurs inτ and is not bound is said to be
free in σ. We writeFTV(σ) for the free type variables ofσ. If FTV(τ) = ∅, thenτ is said to be a
monotype. A type environmentis a finite map from variables to type schemes

TE ∈ TYENV = VAR
fin→ TYSCHEME

It is also useful to view a type environment as a finite set ofassumptionsabout the types of variables.
The set offree type variablesof a type environmentTE is defined to be

FTV(TE) =
⋃

σ∈rng(TE)

FTV(σ)



x ∈ dom(TE) TE(x) � τ

TE ` x : τ

TE± {x 7→ τ ′} ` e : τ

TE ` λ x.e : (τ ′ → τ)

TE ` e1 : (τ ′ → τ) TE ` e2 : τ ′

TE ` (e1 e2) : τ

TE ` e1 : τ ′ TE± {x 7→ CLOSTE(τ ′)} ` e2 : τ

TE ` let x = e1 in e2 : τ

Figure 1: Type inference rules

Theclosure, with respect to a type environmentTE, of a typeτ is defined as

CLOSTE(τ) = ∀α1 · · ·αn.τ

where{α1, . . . , αn} = FTV(τ) \ FTV(TE).

A substitutionis a map from type variables to types. A substitutionS can be naturally extended
to map types to types as follows:

Sι = ι
Sα = S(α)

S(τ1 → τ2) = (Sτ1 → Sτ2)

Application of a substitution to a type schema respects bound variables and avoids capture. It is
defined as:

S(∀α1 · · ·αn.τ) = ∀β1, . . . , βn.S(τ [αi 7→ βi])

whereβi 6∈ dom(S) ∪ FTV (rng(S)). Application of a substitutionS to a type environmentTE is
defined asS(TE) = S ◦ TE. A typeτ ′ is aninstanceof a type schemeσ = ∀α1 · · ·αn.τ , written
σ � τ ′, if there exists a finite substitution,S, with dom(S) = {α1, . . . αn} andSτ = τ ′. If σ � τ ′,
then we say thatσ is ageneralizationof τ ′. Some examples are:

∀α.α � τ , for anyτ ∈ TY

∀α, β.(α → β) � (α → α)
∀α, β.(α → β) � (α → int )

The typing system is given as a set of rules from which sentences of the form “TE ` e : τ ” can
be inferred. This sentence is read as “e has the typeτ under the set of typing assumptionsTE.” The
rules are given in Figure 1

It is worth noting that there is exactly one typing rule for each syntactic form; thus, if we have
a proof of TE ` e : τ , for somee, the form of e uniquely specifies which typing rule was the
last applied in the deduction. This is the formulation of [Tof88] and differs from the system of
[DM82], which has judgements that infer type schemas for expressions and rules for instantiating
and generalizing type schemas. A proof of the equivalence of these two systems can be found in
[CDDK86].

This type inference system is decidable; there exists an algorithm, called algorithmW [DM82]
that infers theprincipal type(i.e., most general under the relation�) of an expression. Algorithm
W is both sound and complete with respect to the inference system. See [DM82] or [Tof88] for the
proof details.

3 Algorithm W

The SML code for AlgorithmW is given in Figure 2. This relies on modules to implement types,

fun algW (env, e) = (case e
of (L.Var x) => let

val sigma = E.lookup(env, x)
in

(S.id, T.freshTy sigma)
end

| (L.App(e1, e2)) => let
val (s1, t1) = algW(env, e1)
val (s2, t2) = algW(S.applySubstToEnv(s1, env), e2)
val beta = T.freshTyVar()
val s3 = U.unify(S.applySubstToTy(s2, t1), T.FnTy(t2, beta))
in

(S.compose(s3, S.compose(s2, s1)), S.applySubstToTy(s3, beta))
end

| (L.Abs(x, e’)) => let
val beta = T.freshTyVar()
val (s1, t1) = algW(E.insert(env, x, T.TyScheme([], beta)), e’)
in

(s1, T.FnTy(S.applySubstToTy(s1, beta), t1))
end

| (L.Let(x, e1, e2)) => let
val (s1, t1) = algW(env, e1)
val xTy = T.closeTy(

E.freeVarsOfEnv(S.applySubstToEnv(s1, env)),
t1)

val (s2, t2) = algW(E.insert(env, x, xTy), e2)
in

(S.compose(s2, s1), t2)
end

(* end case *))

Figure 2: Algorithm W

environments, substitutions, and unification. The unification algorithm, which is owed to Alan
Robinson, is given in Figure 3. Theunify function returns themost general unifier. By this, we
mean that ifunify( τ , τ ′) returnsS, and ifR unifiesτ andτ ′ (i.e.,R(τ) = R(τ ′)), then there
is a substitutionR′, such thatS = R′ ◦ R. The functionoccurs is used for what is called the
“occurs check.” Since recursive types are not allowed, one cannot unify a type variable with a type
that contains it. The occurs check detects this situation and avoids a possible infinite loop.

fun occurs (v1, T.VarTy v2) = (v1 = v2)
| occurs (v, T.BaseTy _) = false
| occurs (v, T.FnTy(ty1, ty2)) = occurs(v, ty1) orelse occurs(v, ty2)

fun unify (T.VarTy v1, ty2 as (T.VarTy v2)) =
if (v1 = v2) then S.id else S.singleton(v1, ty2)

| unify (T.VarTy v1, ty2) =
if (occurs(v1, ty2)) then raise Unify else S.singleton(v1, ty2)

| unify (ty1, T.VarTy v2) =
if (occurs(v2, ty1)) then raise Unify else S.singleton(v2, ty1)

| unify (T.BaseTy a, T.BaseTy b) =
if (a = b) then S.id else raise Unify

| unify (T.FnTy(ty1, ty1’), T.FnTy(ty2, ty2’)) = let
val s1 = unify (ty1, ty2)
val s2 = unify (S.applySubstToTy(s1, ty1’), S.applySubstToTy(s1, ty2’))
in

S.compose(s2, s1)
end

| unify _ = raise Unify

Figure 3: Robinson’s unification algorithm

4 A better algorithm

Algorithm W is not a practical algorithm. It suffers from two sources of inefficiency: the use of ex-
plicit substitutions and the need to determine the free variables in the environment when computing
the type closure. Most real ML compilers use an algorithm that avoids these costs. In this algorithm,
type variables are destructively updated during unification, which avoids the need for substitutions,
and theλ-binding depth of variables is remembered as a way to detect which areλ-bound.

In this algorithm, type variables are represented as a record consisting of an unique ID, and
a updatable kind (see Figure 4). The kind of a type variable starts out asUNIV, and is changed
to INSTANCE, when the variable is unified to a type. The integer argument toUNIV is theλ-
binding depth of the variable. The destructive unification algorithm is given in Figure 5. Note that
when anINSTANCEtype variable is encountered, the instance is followed. Also note that when
a type variable is unified with a type, the depth of the type is adjusted to the minimum depth; this
corresponds to applying the substitution to the variables in the type environment in algorithmW.

The typechecking algorithm is given in Figure 6. Like AlgorithmW, it recursively walks the
term being checked, but it takes an extradepthargument and does not return a substitution. The
pruneTy function is used to prune instantiated type variables.

datatype tyvar = TYVAR of {
id : int,
kind : tvar_kind ref

}

and tvar_kind
= INSTANCE of ty
| UNIV of int

and ty
= VarTy of tyvar
| BaseTy of string
| FnTy of (ty * ty)

and ty_scheme
= TyScheme of (tyvar list * ty)

Figure 4: Representation of types

fun unify (T.VarTy v1, T.VarTy v2) =
if (v1 = v2) then () else unifyVars(v1, v2)

| unify (T.VarTy v1, ty2) = unifyVarWithTy(v1, ty2)
| unify (ty1, T.VarTy v2) = unifyVarWithTy(v2, ty1)
| unify (T.BaseTy a, T.BaseTy b) =

if (a = b) then () else raise Unify
| unify (T.FnTy(ty1, ty1’), T.FnTy(ty2, ty2’)) = (

unify(ty1, ty2);
unify(ty1’, ty2’))

| unify _ = raise Unify

and unifyVars (v1 as T.TYVARkind=k1, ..., v2 as T.TYVARkind=k2, ...) = (
case (!k1, !k2)

of (T.INSTANCE ty1, _) => unifyVarWithTy(v2, ty1)
| (_, T.INSTANCE ty2) => unifyVarWithTy(v1, ty2)
| (T.UNIV d1, T.UNIV d2) => if (d1 < d2)

then k2 := T.INSTANCE(T.VarTy v1)
else k1 := T.INSTANCE(T.VarTy v2)

(* end case *))

and unifyVarWithTy (v as T.TYVARkind, ..., ty) = (case !kind
of (T.INSTANCE ty’) => unify (ty’, ty)

| (T.UNIV d1) => if (occursIn(v, ty))
then raise Unify
else let

val d2 = T.minDepth ty
in

if (d1 < d2) then T.adjustDepth(ty, d1) else ();
kind := T.INSTANCE ty

end
(* end case *))

Figure 5: Destructive unification

fun check (env, depth, e) = (case e
of (L.Var x) => T.freshTy(E.lookup(env, x))

| (L.App(e1, e2)) => let
val ty1 = check (env, depth, e1)
val ty2 = check (env, depth, e2)
val beta = T.freshTyVar depth
in

U.unify (ty1, T.FnTy(ty2, beta));
T.pruneTy beta

end
| (L.Abs(x, e’)) => let

val beta = T.freshTyVar (depth+1)
val ty = check(

E.insert(env, x, T.TyScheme([], beta)),
depth+1, e’)

in
T.FnTy(T.pruneTy beta, ty)

end
| (L.Let(x, e1, e2)) => let

val ty1 = check (env, depth, e1)
val xTy = T.closeTy(depth, ty1)
in

check (E.insert(env, x, xTy), depth, e2)
end

(* end case *))

fun typecheck (env, e) = T.pruneTy(check(env, 0, e))

Figure 6: Typechecking with destructive unification

